Green's function solution of Heat Equation 01-11-11

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Solving heat equation using Green’s function approach

This notebook is a revised version of early work — *Green' s function - Heat equation 02 - 05 - 04. nb*

Define the basic PDE.

```
w[1] = D[G[x, t], t] - κ D[G[x, t], {x, 2}] == DiracDelta[t] DiracDelta[x - x0]
```

$G^{(0,1)}[x, t] - \kappa G^{(2,0)}[x, t] == \text{DiracDelta}[t]\, \text{DiracDelta}[x - x0]$

Easier to work with an expression than with an equation

```
w[2] = w[1] /. Equal → Subtract
```

$-\text{DiracDelta}[t]\, \text{DiracDelta}[x - x0] + G^{(0,1)}[x, t] - \kappa G^{(2,0)}[x, t]$

Laplace transform with respect to time

```
w[3] = LaplaceTransform[w[2], t, ω] /. LaplaceTransform[G[x, t], t, ω] → Ĝ[x, ω] /.
   LaplaceTransform[G^(2,0)[x, t], t, ω] → D[Ĝ[x, ω], {x, 2}] /. G[x, 0] → 0
```

$-\text{DiracDelta}[x - x0] + \omega \hat{G}[x, \omega] - \kappa \hat{G}^{(2,0)}[x, \omega]$

Fourier transform with respect to space

```
w[4] = (FourierTransform[#, x, k]) & /@ w[3]
```

$$-\frac{e^{i\,k\,x0}}{\sqrt{2\pi}} + k^2\,\kappa\,\text{FourierTransform}\left[\hat{G}[x,\omega], x, k\right] + \text{FourierTransform}\left[\omega\,\hat{G}[x,\omega], x, k\right]$$

This has to be manipulated to remove the ω from FourierTransform. I use a rule

```
w[5] = w[4] /.
  FourierTransform[a_. Ĝ[x, ω], x, k] /; FreeQ[a, x] → a FourierTransform[Ĝ[x, ω], x, k]
```

$$-\frac{e^{i\,k\,x0}}{\sqrt{2\pi}} + k^2\,\kappa\,\text{FourierTransform}\left[\hat{G}[x,\omega], x, k\right] + \omega\,\text{FourierTransform}\left[\hat{G}[x,\omega], x, k\right]$$

```
w[6] = w[5] == 0 /. FourierTransform[Ĝ[x, ω], x, k] → G̃[k, ω]
```

$$-\frac{e^{i\,k\,x0}}{\sqrt{2\pi}} + k^2\,\kappa\,\tilde{G}[k,\omega] + \omega\,\tilde{G}[k,\omega] == 0$$

Solve the transformed dependent variable.

```
w[7] = Solve[w[6] ,  G̃[k, ω]][[1, 1]]
```

$$\tilde{G}[k,\omega] \to \frac{e^{i\,k\,x0}}{\sqrt{2\pi}\,\left(k^2\,\kappa + \omega\right)}$$

First, invert the Laplace transform

```
w[8] = (InverseLaplaceTransform[#, ω, t]) & /@ w[7] /.
  InverseLaplaceTransform[G̃[k, ω], ω, t] → Ḡ[k, t]
```

$$\bar{G}[k,t] \to \frac{e^{i\,k\,x0 - k^2\,t\,\kappa}}{\sqrt{2\pi}}$$

where I'm sort of running short of overscripts for notation.

Then, invert the Fourier transform

```
w[9] = (InverseFourierTransform[#, k, x]) & /@ w[8] /.
  InverseFourierTransform[Ḡ[k, t], k, x] → G[x, t] /. Rule → Equal
```

$$G[x,t] == \frac{e^{-\frac{(x-x0)^2}{4\,t\,\kappa}}\sqrt{\frac{1}{t\,\kappa}}}{2\sqrt{\pi}}$$

and we have the Green's function solution of the original pde